Question: I'm trying to generate a clock similar to as in the question asked [here](https://tex.stackexchange.com/questions/49427/reproducing-an-analog-clock-similar-to-the-one-provided-by-the-ifsym-package)

But I want the face to be numbered. Is this possible with this package or any other?

Answer: 
With PSTricks.

```
\documentclass[pstricks,border=12pt]{standalone}
\usepackage{multido}
\SpecialCoor
\begin{document}
\begin{pspicture}(-3,-3)(3,3)
    \pscircle{3}
    \multido{\ia=60+-30,\in=1+1}{12}{\psline(2.9;\ia)(3;\ia)\uput{2.4}[\ia](0,0){\in}}
    \multido{\ia=90+90}{4}{\psline(2.8;\ia)(2.9;\ia)}
    \psline[linewidth=3\pslinewidth]{<->}(2.1;90)(0,0)(1.5;0)
\end{pspicture}
\end{document}

```

![enter image description here](https://i.stack.imgur.com/bIzJN.png)

A clock with exhausted batteries
--------------------------------

```
\documentclass[pstricks,border=12pt]{standalone}
\usepackage{multido}
\SpecialCoor
\begin{document}
\multido{\is=90+-6,\ic=0+1}{60}{%
\begin{pspicture}(-3,-3)(3,3)
    \pscircle{3}
    \multido{\ia=60+-30,\in=1+1}{12}{\psline(2.9;\ia)(3;\ia)\uput{2.4}[\ia](0,0){\in}}
    \multido{\ia=90+90}{4}{\psline(2.8;\ia)(2.9;\ia)}
    \pstVerb{/angle {\ic\space 2 mod 0 gt {90}{105} ifelse} bind def}%
    \psline[linewidth=3\pslinewidth]{<->}(!2.1 angle PtoC)(0,0)(!1.5 angle 90 sub PtoC)
    \psset{linecolor=red}
    \psline{->}(0,0)(2.2;\is)
    \pscircle*{2pt}
\end{pspicture}}
\end{document}

```

![enter image description here](https://i.stack.imgur.com/oQUda.gif)
With Metapost. This is more complicated than it has to be, because it takes care to mark off each minute, and to use different widths and lengths for five-minute ticks as opposed to one-minute ticks, but it's still pretty simple:

```
beginfig(1);

diameter=160;
radius=diameter/2;

pickup pencircle scaled 0.5;

z0 = (radius,radius);
z50 = (radius,diameter);

pair r[];
r[12]  = z50 shifted (0,-radius/4.5);
for i=12 downto 1:
    r[i] = r[i+1] rotatedaround (z0,30);
    z[i] = r[i];
endfor
defaultfont := "cmr17";
defaultscale := 1.5;
label("12",z12);
label("6",z6);
label("9",z9);
label("3",z3);
label("1",z1);
label("5",z5);
label("7",z7);
label("11",z11 shifted (0,-1pt));
label("2",z2);
label("4",z4);
label("8",z8);
label("10",z10 shifted (1pt,0));

pair p[];
pair q[];
numeric circangle;
circangle := 0;
p[0] = z50;
for i=0 step 1 until 60:
    p[i] = p[i-1] rotatedaround (z0,6);
    if (i mod 5 = 0):
        q[i] = 1/12[p[i],z0];
        pickup pencircle scaled 2;
        draw p[i]--q[i];
        pickup pencircle scaled 0.5;
    else:
        q[i] = 1/30[p[i],z0];
        pickup pencircle scaled 0.5;
        draw p[i]--q[i];
    fi
    draw p[i-1]..p[i];
endfor

endfig;
end;

```

I've always been partial to Metapost, as I find it much easier to program than tikZ. tikZ does have the benefit of being able to be thrown into macros, though, as Harish Kumar's excellent answer above demonstrates.

I know that, strictly speaking, this isn't a TeX answer; but Metapost has such a pedigree among TeXnicians (even Knuth himself uses it for his figures, I'm told) that it seemed a fair answer to me.